1
5

7. КОНЦЕПЦИИ МОДУЛЬНОГО ПРОГРАММИРОВАНИЯ

Модуль – это автономно, то есть отдельно от программы, транслируемая  программная единица, которая может состоять из набора подпрограмм и данных и подключается на этапе формирования выполняемой программы, то есть файла prog.exe.

Наличие модулей позволяет:

1) Программировать и отлаживать программу по частям.

2) Создавать собственные библиотеки подпрограмм и данных.

3) Пользоваться библиотеками стандартных подпрограмм – стандартными модулями.

4) Существенно увеличивать полный объём программы, добавляя к сегменту кода основной программы до 10 сегментов модулей.

7.1. СТРУКТУРА МОДУЛЯ

Модуль должен содержать: заголовок, интерфейсную часть и  исполнительную часть,  может  содержать секцию инициализации.

Заголовок модуля содержит имя модуля, которое должно быть уникально:

Unit Stack;

Модуль должен быть помещен в файл, имя которого совпадает с именем модуля и имеет расширение .pas:  stack.pas . Модуль транслируется отдельно. После трансляции получается файл stack.tpu в текущем каталоге.

 Интерфейсная часть может содержать описания констант, типов, переменных и полные заголовки подпрограмм, которые используются основной программой.
Исполнительная часть  содержит описание всех подпрограмм модуля.

Секция инициализации  может содержать  начальные действия, необходимые сразу после запуска программы, например: установка начальных значений переменных, установка связи с файлами и др.

Пример модуля:
UNIT Stack;  { Модуль для работы со стеком сохранить в файле stack.pas ! }

 INTERFACE { Интерфейсная часть модуля }

Type Tinf = Char;   { Тип информации элемента }

TP = ^Telem;   { Тип cсылка на элемент стека  }

Telem = Record   { Тип элемента стека  - запись  }

    inf:Tinf; { Поле информации }

    pSled:TP  { Поле ссылки на следующий элемент }

    End;

Procedure InitStack(Var top:TP);          {Инициализация стека}

Function EmptyStack(top:TP):Boolean;    {Проверка пустоты стека}

Procedure InStack(infEl:Tinf; Var top:TP); {Включение элемента}

Procedure OutStack(Var infEl:Tinf;Var top:TP);{Извлечение элемента}

IMPLEMENTATION   { Исполнительная часть модуля }

Procedure InitStack(Var top:TP);


{ Процедура инициализации стека }

Begin
top:=NIL { Ссылка на вершину стека } 

End; { InitStack }

Function EmptyStack(top:TP):Boolean; 

{ Функция проверки пустоты стека }

Begin

EmptyStack:= top=NIL

End; { EmptyStack }

Procedure InStack(infEl:Tinf; Var top:TP);

{ Процедура включения элемента в стек }

Var pn:TP;    { Ссылка на  новый элемент }

Begin

New(pn);  { Резервирование памяти для нового элемента } 

pn^.inf:=infEl; { Заполнение поля информации } 

pn^.pSled:=top; { Заполнение поля ссылки } 
top:=pn         { Новая ссылка на вершину стека }

End; { InStack }

Procedure OutStack(Var infEl:Tinf; Var top:TP);

{ Процедура извлечения элемента из стека }

Var pu:TP; { Ссылка на удаляемый элемент }

Begin

If top<>NIL { Если стек не пуст  }

Then Begin

infEl:=top^.inf;  { Извлечение информации }

pu:=top;    { Запоминание ссылки на удаляемый элемент } 

top:=top^.pSled;  { Новая ссылка на вершину стека } 

Dispose(pu)       { Удаление элемента стека }

End

End; { OutStack }

END.

7.2. УСТАНОВКА СВЯЗИ С МОДУЛЕМ

Связь устанавливается описательным предложением: Uses <Список имен модулей>;

Например, описание в основной программе:

Uses CRT,Stack;

позволяет ей использовать все имена, указанные в интерфейсных частях этих модулей.

Пример использования модуля:

Program Control;
{ Пример 21 - 2007 }

{ Программа контроля правильности расстановки скобок в программах }

Uses Stack;    { Подключение модуля для работы со стеком }

Procedure Brackets(Const nameF:String);

{ Процедура проверки правильности расстановки скобок

в тексте программы - текстовом файле с именем nameF }

Procedure Error(n:Byte;s:Char);

{ Процедура расшифровки ошибок }

Begin

Case n Of

1: WriteLn('Нет соответствия ',s,' скобок');

2: WriteLn('Лишняя закрывающая скобка ',s);

3: Writeln('Лишняя открывающая скобка ',s)

End;

End; { Error }

Var FT:Text;  top:TP;  s,sz,sk:Char;

Begin

Assign(FT,nameF); Reset(FT);

InitStack(top);
{ Инициализация стека }

While Not Eof(FT) Do    { Пока не конец файла }

Begin While Not EoLn(FT) Do { Пока не конец строки }

 
     Begin Read(FT,s); { Чтение символа из файла }




Case s Of




'(','[','{':InStack(s,top);{Поместить скобку в стек}




'}',']',')':If Not EmptyStack(top){Если стек не пуст}





   Then Begin { Стек не пуст }







      OutStack(sk,top);{Извлечь из стека }







      Case sk Of






     
'(': sz:=')';






     
'[': sz:=']';






     
'{': sz:='}'






         End;






         If s<>sz { Если скобки разных видов }






         Then  Begin  Error(1,s); Exit

  End

      End 
     Else Begin{Стек пуст}Error(2,s); Exit

                              End

  End; { Case }

 End; { While }  ReadLn(FT)
{ Чтение конца строки }

End;

Close(FT);

If EmptyStack(top)
{ Если стек пуст }

Then WriteLn('Скобки расставлены правильно !')

Else Begin OutStack(sk,top); Error(3,sk)

  End

End; { Brackets }

Function FileExist(Const nameF:String):Boolean;

{ Функция проверки существования файла любого вида }

Var F:File; { Любой вид файла }

Begin Assign(F,nameF);

{$I-}  { Отключение автоматического контроля чтения-записи }

Reset(F);  { Открытие файла для чтения }

{$I+}  { Включение автоматического контроля чтения-записи }

If IOResult=0  { Если файл существует }

Then Begin FileExist:=True;

    Close(F)

  End

Else FileExist:=False

End; { FileExist }

Var nameFT:String;  s:Char;

Begin

Repeat

WriteLn('Полное имя файла с текстом проверяемой программы?');

ReadLn(nameFT);

If FileExist(nameFT) Then Brackets(nameFT)

Else WriteLn('Файл ', nameFT,' не существует !');

WriteLn('Проверить еще одну программу ?');

ReadLn(s);

Until Not ( s in ['Д','д','Y','y'] );

End.
7.3. КОМПИЛЯЦИЯ МОДУЛЕЙ

Режимы компиляции: Compile, Make, Build.

Compile: Все модули, объявленные в Uses, должны быть заранее откомпили-рованы и помещены в одноименные файлы с расширением .tpu в каталоге, объявленном опцией UNIT DIRECTORIES.

Make: Компилятор проверяет наличие tpu-файлов для каждого объявленного модуля. Если для какого-либо модуля нет tpu-файла, то система находит pas-файл с исходным текстом и компилирует его. Если дата создания кода модуля более ранняя, чем дата создания текста модуля, то происходит перекомпиляция.

Build: Все существующие tpu-файлы игнорируются и система перекомпилирует все  модули, объявленные в Uses.
7.4. РАЗМЕЩЕНИЕ ПРОГРАММ И ДАННЫХ

В ОПЕРАТИВНОЙ ПАМЯТИ (ОП) КОМПЬЮТЕРА
                                          ОП

	Сегмент кода основной программы

	Сегменты кодов подключаемых модулей

	Сегмент кода модуля System

	Сегмент данных

	Сегмент стека

	Динамическая

память







по 64 К       на каждый из перечисленных 


        в Uses  в обратном порядке








Подключается автоматически





  < 300 K       Куча





64 К        Статическая память


   хранилище констант и переменных программы





16 К   	   Системный стек 


   используется  при вызове подпрограмм





64 К          





Максимум Мбайт





1 Мбайт





64 К       Program---End.








